NAG Fortran Library Routine Document

S14AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S14AGF returns the value of the logarithm of the Gamma function $\ln \Gamma(z)$ for complex z, via the routine name.

2 Specification

```
complex*16 FUNCTION S14AGF (Z, IFAIL)
INTEGER                      IFAIL
complex*16                   Z
```

3 Description

S14AGF evaluates an approximation to the logarithm of the Gamma function $\ln \Gamma(z)$ defined for $\mathrm{Re}(z) > 0$ by

$$\ln \Gamma(z) = \ln \int_0^\infty e^{-t} t^{z-1} dt$$

where $z = x + iy$ is complex. It is extended to the rest of the complex plane by analytic continuation unless $y = 0$, in which case z is real and each of the points $z = 0, -1, -2, \ldots$ is a singularity and a branch point.

S14AGF is based on the method proposed by Kölbig (1972) in which the value of $\ln \Gamma(z)$ is computed in the different regions of the z plane by means of the formulae

$$\begin{aligned} \ln \Gamma(z) &= \left(z - \tfrac{1}{2}\right) \ln z - z + \tfrac{1}{2} \ln 2\pi + z \sum_{k=1}^{K} \frac{B_{2k}}{2k(2k-1)} z^{-2k} + R_K(z) && \text{if } x \geq x_0 \geq 0, \\ &= \ln \Gamma(z+n) - \ln \prod_{\nu=0}^{n-1} (z+\nu) && \text{if } x_0 > x \geq 0, \\ &= \ln \pi - \ln \Gamma(1-z) - \ln(\sin \pi z) && \text{if } x < 0, \end{aligned}$$

where $n = [x_0] - [x]$, $\{B_{2k}\}$ are Bernoulli numbers (see Abramowitz and Stegun (1972)) and $[x]$ is the largest integer $\leq x$. Note that care is taken to ensure that the imaginary part is computed correctly, and not merely modulo 2π.

The routine uses the values $K = 10$ and $x_0 = 7$. The remainder term $R_K(z)$ is discussed in Section 7.

To obtain the value of $\ln \Gamma(z)$ when z is real and positive, S14ABF can be used.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Kölbig K S (1972) Programs for computing the logarithm of the gamma function, and the digamma function, for complex arguments *Comp. Phys. Comm.* **4** 221–226

5 Parameters

1: Z – ***complex*16*** *Input*

On entry: the argument z of the function.

Constraint: Re(Z) must not be ‘too close’ (see Section 6) to a non-positive integer when Im(Z) = 0.0.

2: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, Re(Z) is ‘too close’ to a non-positive integer when Im(Z) = 0.0. That is, ABS(Re(Z) – NINT(Re(Z))) < ***machine precision*** × NINT(ABS(Re(Z))).

7 Accuracy

The remainder term $R_K(z)$ satisfies the following error bound:

$$\begin{aligned} |R_K(z)| &\leq \frac{|B_{2K}|}{|(2K-1)|} z^{1-2K} \\ &\leq \frac{|B_{2K}|}{|(2K-1)|} x^{1-2K} \text{if } x \geq 0. \end{aligned}$$

Thus $|R_{10}(7)| < 2.5 \times 10^{-15}$ and hence in theory the routine is capable of achieving an accuracy of approximately 15 significant digits.

8 Further Comments

None.

9 Example

The example program evaluates the logarithm of the Gamma function $\ln \Gamma(z)$ at $z = -1.5 + 2.5i$, and prints the results.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     S14AGF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      COMPLEX *16      Y, Z
      INTEGER          IFAIL
*     .. External Functions ..
      COMPLEX *16      S14AGF
      EXTERNAL         S14AGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'S14AGF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) '       Z                 ln(Gamma(Z))',
     +  '           IFAIL'
      WRITE (NOUT,*)
   20 READ (NIN,*,END=40) Z
      IFAIL = 0
*
      Y = S14AGF(Z,IFAIL)
*
      WRITE (NOUT,99999) Z, Y, IFAIL
      GO TO 20
   40 STOP
*
99999 FORMAT (1X,'(',F5.1,',',F5.1,' )   (',1P,D12.4,',',D12.4,' )',I7)
      END
```

9.2 Program Data

```
S14AGF Example Program Data
 (-1.5, 2.5)  : Value of Z
```

9.3 Program Results

```
 S14AGF Example Program Results

        Z                 ln(Gamma(Z))            IFAIL

 ( -1.5,  2.5 )   ( -5.0140D+00, -4.0718D+00 )      0
```